1.) Wiederholung: Stetigkeit (Definition und elementare Eigenschaften), Anwendungen von Stetigkeit (Zwischenwertsatz, Intervallschachtelung, Extrema, Grenzwerte und Stetigkeit), Beispiele und erste Theorie von Vektorräumen, Faktorräume, Restklassen und Homomorphiesatz

Wiederhole die obigen Themen, wir stellen Fragen im Testat.

2.) Matrix einer linearen Abbildung

Aufgaben: 7

```
> restart;
  with(LinearAlgebra):
```

Definitionen

MATH: Seien V und W Vektorräume über dem Körper K mit Basen $B = (b_1, \ldots, b_n) \in V^n$ und $C = (c_1, \ldots, c_m) \in W^m$. Wir wollen eine bequeme Art finden, eine lineare Abbildung

$$f : V \to W$$

zu beschreiben. Bekanntlich haben wir Isomorphismen

$$\kappa_B : V \to K^n : v \mapsto v_B,$$

$$\kappa_C : W \to K^m : w \mapsto w_C,$$

wobei v_B, w_C die Koordinatenvektoren von *v* und *w* bezüglich der Basis B bzw. C sind.

Also ist die Komposition

$$\kappa_C \circ f \circ \kappa_B^{-1} : K^n \to K^m$$

auch eine lineare Abbildung. Jede lineare Abbildung $K^n \to K^m$ wird aber durch eine eindeutig bestimmte Matrix aus $K^{m \times n}$ beschrieben. Wir nennen diese Matrix

$$A^f_{C,B} \in K^{m \times n},$$

die **darstellende Matrix von f bezüglich der beiden Basen B und C**.

Es gilt dann:

$$A^f_{C,B} \cdot v_B = (f(v))_C.$$

Mit anderen Worten: Die i-te Spalte von $A^f_{C,B}$ ist die Koordinatenspalte von $f(b_i)$ bezüglich der Basis C von W.

Beispiel/Definition: $V = W$ und $f = \mathrm{id} :=$ Identität von V. Dann heißt $A^{id}_{C,B}$ die Basiswechselmatrix von B nach C. Multipliziert man $A^{id}_{C,B}$ an die

Koordinatenspalte eines Vektors bezüglich B so erhält man die Koordinatenspalte bezüglich C.

DENKANSTOSS: Interpretiere und beweise die Formel:

$$C \cdot A_{C,B}^{id} = B$$

oder die allgemeinere Formel

$$C \cdot A_{C,B}^{f} = f(B).$$

Hinweis:

$$B \cdot v_B = v$$

für alle $v \in V$.

MATH: Jede invertierbare Matrix $A \in K^{n \times n}$ lässt sich als Basiswechselmatrix interpretieren, sagen wir von einer Basis B zu einer Basis C. Sind zwei der drei Daten A, B, C gegeben, so lässt sich die dritte eindeutig bestimmen. Ist C z. B. die Standardbasis von K^n, so besteht B aus den Spaltenvektoren von $A = A_{C,B}^{id}$.

DENKANSTOSS: Wie steht es mit den beiden anderen Möglichkeiten, d.h. wo C, B bzw. A, B gegeben sind?

Beispiel:

```
> C:=map(i->x^i,[$0..5]);
```

$$C := [1, x, x^2, x^3, x^4, x^5] \quad \textbf{(2.1.1)}$$

```
> B:=map(i->(1+x)^i,[$0..5]);
```

$$B := [1, 1+x, (1+x)^2, (1+x)^3, (1+x)^4, (1+x)^5] \quad \textbf{(2.1.2)}$$

sind beides Basen von $V := K[x]_{\leq 5}$. Wir betrachen die Abbildung $\mathrm{f} := \mathrm{id} :=$ Identität von V. Die Matrix $A_{C,B}^{id}$ ist also

```
> A_CB:=Matrix(6,6,(i,j)->coeff(B[j],x,i-1));
```

$$A\_CB := \begin{bmatrix} 1 & 1 & 1 & 1 & 1 & 1 \\ 0 & 1 & 2 & 3 & 4 & 5 \\ 0 & 0 & 1 & 3 & 6 & 10 \\ 0 & 0 & 0 & 1 & 4 & 10 \\ 0 & 0 & 0 & 0 & 1 & 5 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \quad \textbf{(2.1.3)}$$

Wir fragen nach $A_{B,C}^{id}$.

MATH: Ist $\mathrm{g} : W \rightarrow X$ eine weitere lineare Abbildung und $\mathrm{D} = (d_1, \ldots, d_l)$ eine Basis von X, so gilt:

$$A_{\mathrm{D},B}^{g \circ f} = A_{\mathrm{D},C}^{g} \cdot A_{C,B}^{f}$$

wobei also links die Matrix der Komposition zweier linearer Abbildungen steht und rechts das Produkt der Matrizen dieser beiden linearen Abbildungen.

BEISPIEL (Fortsetzung von oben): $V = W = X$, $g = \mathrm{id}$ und $D = B$. Da offenbar

$A^{id}_{B,\,B} = A^{id}_{C,\,C} =$ Einheitsmatrix,

bekommen wir:

$A^{id}_{B,\,C} = \left(A^{id}_{C,\,B}\right)^{-1}$.

```
> A_CB^(-1);
```

$$\begin{bmatrix} 1 & -1 & 1 & -1 & 1 & -1 \\ 0 & 1 & -2 & 3 & -4 & 5 \\ 0 & 0 & 1 & -3 & 6 & -10 \\ 0 & 0 & 0 & 1 & -4 & 10 \\ 0 & 0 & 0 & 0 & 1 & -5 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \tag{2.1.4}$$

ÜBUNG [01]:

Sei C wie oben und D die unten angegebene Basis von $K[x]_{\leq 5}$.

1) Finde die Matrizen

$A^{id}_{C,\,\mathrm{D}}$ und $A^{id}_{\mathrm{D},\,C}$.

2) Wie können diese uns helfen, einen Vektor aus einer Basis in die andere umzurechnen?

3) Stelle das Polynom $2 \cdot x^2 + 3$ in der Basis C dar und rechne es mit Hilfe der obigen Matrizen in die Basis D um.

```
> C;
  unprotect(D):
  D:=map(i->(2+x)^i,[$0..5]);
```

$$[1,\, x,\, x^2,\, x^3,\, x^4,\, x^5]$$

$$\mathrm{D} := [1,\, 2 + x,\, (2 + x)^2,\, (2 + x)^3,\, (2 + x)^4,\, (2 + x)^5] \tag{2.1.5}$$

MATH: Ist $A \in K^{m \times n}$ eine Matrix über dem Körper K, so heißt

$Z(A) := \langle A_{1,-}, A_{2,-}, \ldots, A_{m,-} \rangle$ der **Zeilenraum** von A und seine Dimension

$ZRg(A) := Dim(Z(A))$ der **Zeilenrang** von A. Analog dazu heißt

$S(A) := \langle A_{-,1}, A_{-,2}, \ldots, A_{-,n} \rangle$ der **Spaltenraum** von A und seine Dimension

$SRg(A) := Dim(S(A))$ der **Spaltenrang** von A.

MATH: Ist $A \in K^{m \times n}$ eine Matrix über dem Körper K und bezeichnet SZF_A die Matrix der strikten Zeilenstufenform von A, so gilt: $Z(A) = Z(SZF_A)$.

Beweis:

1) $Z(A) \subseteq Z(SZF_A)$

Es gilt: $G \cdot SZF_A = A$ nach Gaußalgorithmus für eine invertierbare Matrix $G \in K^{m \times m}$.

Sei nun $v := \left(\sum_{i=1}^{m} a_i \cdot A_{i,-}\right) \in Z(A)$, also $a := (a_1, \dots, a_m) \in K^{1 \times m}$ mit $a \cdot A = v$.
Dann für $a' := a \cdot G$:
$a' \cdot SZF_A = a \cdot G \cdot SZF_A = a \cdot A = v$
Also liefert uns a' die gesuchte Linearkombination der Zeilen von SZF_A :
$v = \left(\sum_{i=1}^{m} a'_i \cdot (SZF_A)_{i,-}\right)$

2) $Z(A) \supseteq Z(SZF_A)$
Analog.

ÜBUNG [02]:

1) Beweise, dass A und SZF_A denselben Spaltenrang haben.
2) Folgere, dass $ZRg(A) = SRg(A)$ gilt, weshalb $Rg(A) := ZRg(A)$ bzw. $Rg(A) := SRg(A)$ eine sinnvolle Definition für den **Rang** einer Matrix ist.

Lineare (Un)abhängigkeit

Wir wollen sehen, wie man lineare Abbildungen und Matrizen benutzen kann, um lineare Unabhängigkeit nachzuweisen.

Erinnerung: K-wertige Folgen, also Abbildungen $f: \mathbb{N} \to K$ bilden einen K-Vektorraum $K^{\mathbb{N}}$.

```
> (f+g)[i] = f[i]+g[i];
```

$$(f+g)_i = f_i + g_i \tag{2.2.1}$$

definiert die Summe der Folgen f und g und für $k \in K$ definiert

```
> (k*f)[i] = k*f[i];
```

$$(k f)_i = k f_i \tag{2.2.2}$$

die Folge $k \cdot f$.

MATH: Das Einschränken von Folgen auf eine Teilmenge I von $\mathbb{N}$ ist eine lineare Abbildung. (**DENKANSTOSS**: Warum?)

Beispiel: $V := \langle a_0, a_1, a_2 \rangle \subseteq \mathbb{Q}^{\mathbb{N}}$:

```
> a0 := i -> 1;
  a1 := i -> i;
  a2 := i -> i^2;
```

$$a0 := i \to 1$$
$$a1 := i \to i$$
$$a2 := i \to i^2 \tag{2.2.3}$$

```
> B:=[a0,a1,a2];
```

$$B := [a0, a1, a2] \quad (2.2.4)$$

Wir betrachten die Restriktion der Folgen auf die Menge $\{1, \ldots, 4\}$:

```
> res:=map(a->map(a, [$1..4]),B);
```

$$res := [[1, 1, 1, 1], [1, 2, 3, 4], [1, 4, 9, 16]] \quad (2.2.5)$$

Auf dem Raum $\mathbb{Q}^4$ der Folgen der Länge 4 haben wir die Standardbasis:

```
> map(i->map(j->if i=j then 1 else 0 end if,[$1..4]),[$1..4]);
```

$$[[1, 0, 0, 0], [0, 1, 0, 0], [0, 0, 1, 0], [0, 0, 0, 1]] \quad (2.2.6)$$

Die Koordinatenspalten der Elemente von **res** bezüglich dieser Basis sind gerade die Spalten von

```
> resK:=Matrix(4,3,(i,j)->res[j][i]);
```

$$resK := \begin{bmatrix} 1 & 1 & 1 \\ 1 & 2 & 4 \\ 1 & 3 & 9 \\ 1 & 4 & 16 \end{bmatrix} \quad (2.2.7)$$

Diese Spalten sind linear unabhängig über $\mathbb{Q}$, denn

```
> NullSpace(resK);
```

$$\{\,\} \quad (2.2.8)$$

MAPLE: Der Befehl **NullSpace** bestimmt den Kern einer Matrix mittels *Gauß*-Algorithmus, welcher uns schon bekannt ist.

ÜBUNG [03]:

Verstehe das obige Beispiel, welches die lineare Unabhängigkeit im Bild nutzt, und zeige: Dadurch folgt: B ist linear unabhängig über $\mathbb{Q}$.
Verstehe insbesondere welche lineare Abbildung betrachtet wurde und wieso dieses Verfahren für beliebige lineare Abbildungen funktioniert.

MATH: Wir wollen die gerade die gerade entwickelte Idee benutzen, um mögliche lineare Abhängigkeiten für

$$X := \left(i \mapsto 1, i \mapsto i, i \mapsto i^2, i \mapsto i^3, i \mapsto \sum_{j=1}^{i} j^2\right)$$

in $\mathbb{Q}^{\mathbb{N}}$ zu finden.

```
> a3:= i -> i^3;
  a4:= i -> add(j^2,j=1..i);
```

$$a3 := i \to i^3$$

$$a4 := i \to add(j^2, j = 1..i) \quad (2.2.9)$$

```
> X:=[a0,a1,a2,a3,a4];
```

$$X := [a0, a1, a2, a3, a4] \quad (2.2.10)$$

Wir schränken auf $\{1, \ldots, n\}$ ein:

```
> n:=7;
```

$$n := 7 \quad (2.2.11)$$

```
> Xres:=map(r->map(i->r(i),[$1..n]),X);
```

$$Xres := [[1, 1, 1, 1, 1, 1, 1], [1, 2, 3, 4, 5, 6, 7], [1, 4, 9, 16, 25, 36, 49], [1, 8, 27, 64, 125, 216, 343], [1, 5, 14, 30, 55, 91, 140]] \tag{2.2.12}$$

```
> A:=Matrix(n,5,(i,j)->Xres[j][i]);
```

$$A := \begin{bmatrix} 1 & 1 & 1 & 1 & 1 \\ 1 & 2 & 4 & 8 & 5 \\ 1 & 3 & 9 & 27 & 14 \\ 1 & 4 & 16 & 64 & 30 \\ 1 & 5 & 25 & 125 & 55 \\ 1 & 6 & 36 & 216 & 91 \\ 1 & 7 & 49 & 343 & 140 \end{bmatrix} \tag{2.2.13}$$

```
> N:=op(NullSpace(A));
```

$$N := \begin{bmatrix} 0 \\ -\frac{1}{6} \\ -\frac{1}{2} \\ -\frac{1}{3} \\ 1 \end{bmatrix} \tag{2.2.14}$$

Die einzige Möglichkeit für eine lineare Abhängigkeit von *B* steht in dieser Spalte.

```
> map(i-> -1/6*a1(i) - 1/2*a2(i) - 1/3*a3(i) + a4(i), [$1..20]
  ) ;
  map(i->add(N[j]*X[j](i),j=1..nops(X)),[$1..20]);
```

$$[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0]$$

$$[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0] \tag{2.2.15}$$

Jetzt haben wir die moralische Gewissheit, dass wir eine lineare Abhängigkeit haben. Da die Einschränkung auf $\{1, \ldots, n\}$ nicht injektiv war, haben wir aber immer noch keinen Beweis.

Man kann die Weisheit von Maple benutzen

```
> Sum(j^2,j=1..i) = expand(sum(j^2,j=1..i));
```

$$\sum_{j=1}^{i} j^2 = \frac{1}{3} i^3 + \frac{1}{2} i^2 + \frac{1}{6} i \tag{2.2.16}$$

oder den üblichen Induktionsbeweis führen, um diese Gleichheit nachzuweisen.

Für die folgende Aufgaben müssen wir mit großen Matrizen arbeiten und dafür erstmal Maple mit dem nächsten Befehl überreden, solche bis zur Dimension 20x20 anzuzeigen.

```
> interface(rtablesize=20);
```

$$10 \tag{2.2.17}$$

ÜBUNG [04]:

1) Benutze die obige Methode, um Kandidaten für eine geschlossene Formel für

$$\sum_{j=1}^{i} j^{10}$$

als Polynom in i zu finden.

2) Benutze **sum** und **expand**, um diese Formel zu testen.

Beispiel: Arithmetische Progressionen

MATH: Auf dem $\mathbb{Q}^{\mathbb{N}}$ haben wir den Differenzenoperator als lineare Abbildung:

$$\Delta : \mathbb{Q}^{\mathbb{N}} \to \mathbb{Q}^{\mathbb{N}} : (f_1, f_2, f_3, \dots) \mapsto (f_2 - f_1, f_3 - f_2, \dots).$$

Wir schränken auf den von a_0, a_1 erzeugten Teilraum A_1 ein.

```
> [a0,a1];
  map(a->a(i), %);
```

$$[a0, a1]$$

$$[1, i] \tag{2.3.1}$$

Die Elemente von A_1 heißen auch **arithmetische Progressionen.** Dieser Teilraum hat offenbar

$$B := (a_0, a_1) \in A_1^2$$

als Basis. Weiter ist klar/wird spätestens unten klar:

$$\Delta(A_1) \subseteq A_1.$$

Es folgt die Matrix M der Einschränkung von Δ auf A_1 bezüglich des Basenpaares B für den Definitionsbereich und B für den Bildbereich:

```
> M:=<<0 | 1> , <0 | 0>>;
```

$$M := \begin{bmatrix} 0 & 1 \\ 0 & 0 \end{bmatrix} \tag{2.3.2}$$

Die erste Spalte erhält man, da $a_0 = i \mapsto 1 = (1, 1, 1, \dots)$ durch Δ auf $(1-1, 1-1, \dots) = (0, 0, \dots) = 0 \cdot a_0 + 0 \cdot a_1$ abgebildet wird. Analog erhält man die zweite Spalte durch $\Delta(a_1) = (2-1, 3-2, \dots) = (1, 1, \dots) = 1 \cdot a_0 + 0 \cdot a_1$.

ÜBUNG [05]:

1) Die Einschränkung von Δ auf $\langle a_0, a_1, a_2, a_3 \rangle$ liefert eine Abbildung

$$\Delta_{|\langle a_0, a_1, a_2, a_3\rangle}: \langle a_0, a_1, a_2, a_3\rangle \to \langle a_0, a_1, a_2\rangle$$

Bestimme die Matrix dieser Abbildung bezüglich der beiden Basen (a_0, a_1, a_2, a_3) und (a_0, a_1, a_2).

2) Man bestimme alle Elemente aus $\langle a_0, a_1, a_2, a_3\rangle$, die im Kern von Δ liegen.

Beispiel: Differentiation

MATH: Die formale Ableitung δ für Polynome,

$$\delta: \mathbb{Q}[x] \to \mathbb{Q}[x]: \sum_{i=0}^{n} f_i \cdot x^i \mapsto \sum_{i=1}^{n} f_i \cdot i \cdot x^{i-1}$$

definiert eine lineare Abbildung von $\mathbb{Q}[x]$ in sich. Wir schränken auf

$$\mathbb{Q}[x]_{\leq n}$$

ein, also auf den Raum der Polynome vom Grad $\leq n$, und erhalten eine lineare Abbildung

$$\mathbb{Q}[x]_{\leq n} \to \mathbb{Q}[x]_{\leq n-1}.$$

```
> poly:=add(f[i]*x^i,i=0..5);
```

$$poly := x^5 f_5 + x^4 f_4 + x^3 f_3 + x^2 f_2 + x f_1 + f_0 \qquad \textbf{(2.4.1)}$$

```
> diff(poly,x);
```

$$5 x^4 f_5 + 4 x^3 f_4 + 3 x^2 f_3 + 2 x f_2 + f_1 \qquad \textbf{(2.4.2)}$$

ÜBUNG [06]:

1) Gib die Matrix der formalen zweifachen Ableitung

$$\delta^2 = \delta \circ \delta: \quad \mathbb{Q}[x]_{\leq 6} \to \mathbb{Q}[x]_{\leq 4}$$

bezüglich der Basen

```
> B6:=map(i->(x+1)^i,[$0..6]);
```

$$B6 := [1, 1+x, (1+x)^2, (1+x)^3, (1+x)^4, (1+x)^5, (1+x)^6] \qquad \textbf{(2.4.3)}$$

und

```
> B4:=map(i->(x-1)^i,[$0..4]);
```

$$B4 := [1, x-1, (x-1)^2, (x-1)^3, (x-1)^4] \qquad \textbf{(2.4.4)}$$

einmal direkt an und zum anderen als Produkt der Matrizen, die die formalen Ableitungen

$$\mathbb{Q}[x]_{\leq 6} \to \mathbb{Q}[x]_{\leq 5} \qquad \text{und} \qquad \mathbb{Q}[x]_{\leq 5} \to \mathbb{Q}[x]_{\leq 4}$$

beschreiben, wobei als Basis von $\mathbb{Q}[x]_{\leq 5}$ die Standardbasis

```
> B5:=map(i->(x)^i,[$0..5]);
```

$$B5 := [1, x, x^2, x^3, x^4, x^5] \qquad \textbf{(2.4.5)}$$

benutzt wird.

2) Bestimme mit Hilfe der Matrix von δ^2 den Kern von δ^2.